Maple Code for Random Polygons in $\mathbb{R}^2$ and $\mathbb{R}^3$

```
restart;
with(LinearAlgebra): with(Statistics): with(plottools): with(plots):
n := 7: QL[1] := Vector[row]([0, 0]): QM[1] := Vector[row]([0, 0]):

A,B := QRDecomposition(Sample(RandomVariable(Normal(0, 1)), [n, 2])):
CC := Matrix([seq([A[i, 1]^2 - A[i, 2]^2, 2*A[i, 1]*A[i, 2]], i = 1 .. n)]):

for i from 1 to n do  QL[i + 1] := QL[i] + CC[i]  end do:

BL := seq(line(QL[i], QL[i + 1], color = red), i = 1 .. n): display(BL);
CL := seq(line(QL[1], CC[i], color = red), i = 1 .. n):  display(CL);

DL := [seq(argument(CC[i][1] + CC[i][2]*I), i = 1 .. n)]:
PER := sort(DL, 'output = permutation'):

for i to n do QM[i + 1] := QM[i] + CC[PER[i]]  end do:

EL := seq(line(QM[i], QM[i + 1], color = red), i = 1 .. n): display(EL);
```

```
restart;
with(LinearAlgebra):with(Statistics):with(plottools):with(plots):
n := 10: QL[1] := Vector[row]([0, 0, 0]):

A := seq(Transpose(Sample(RandomVariable(Normal(0, 1)), n)), i = 1 .. 4):
B := GramSchmidt([A[1] + A[2]*I, A[3] + A[4]*I], normalized):
C1 := Re(B[1]): C2 := Im(B[1]): C3 := Re(B[2]): C4 := Im(B[2]):

CC := Matrix([seq([C1[i]^2 + C2[i]^2 - C3[i]^2 - C4[i]^2,
  2*(-C1[i]*C4[i] + C2[i]*C3[i]), 2*(C1[i]*C3[i] + C2[i]*C4[i])], i = 1 .. n)]):

for i from 1 to n do  QL[i + 1] := QL[i] + CC[i]  end do:

BL := seq(line(QL[i], QL[i + 1], color = red), i = 1 .. n):  display(BL);
```